


# “Turtle Tuesday” Logo Challenges

---

April 4, 2023

## Challenge No. 27

Can you draw a truck or other vehicle in Logo?



Our solution:

You can add a comment to explain your code by adding text after a semicolon.


```
TO TRUCK
  DRAW SETW 2 PU SETX XCOR - 200 PD      ; move to starting position
  REPEAT 2 [FD 160 RT 90 FD 440 RT 90] ; draw the box
  RT 90 FD 60 WHEEL
  FD 70 WHEEL
  FD 150 WHEEL
  FD 70 WHEEL
  FD 90 LT 90
  ; draw the cab
  FD 160 RT 90 FD 100 RT 90 FD 80
  LT 90 FD 80 RT 90 FD 80
  RT 90 FD 180 BK 110 RT 180
  WHEEL
END

TO WHEEL
  REPEAT 20 [FD 10 RT 18]
END
```

**April 11, 2023**

## Challenge No. 28

Can you draw a city skyline in Logo? Yours can look different!




Our solution:

```
TO SKYLINE
  DRAW PU SETPOS [-250 -250] PD SETW 2
  REPEAT 4 [FD 200 RT 90 FD 60 RT 90]
  RT 90 FD 60 LT 90
  REPEAT 2 [FD 280 RT 90 FD 70 RT 90]
  RT 90 FD 70 LT 90
  REPEAT 2 [FD 360 RT 90 FD 50 RT 90] FD 360 RT 90
  REPEAT 3 [FD 50 LT 120]
  LT 90 BK 360 RT 90 FD 50 LT 90
  REPEAT 2 [FD 80 RT 90 FD 80 RT 90]
  FD 80 REPEAT 2 [FD 25 RT 90 FD 80 RT 90] BK 80
  RT 90 FD 80 LT 90
  REPEAT 2 [FD 300 RT 90 FD 200 RT 90]
  RT 90 FD 200 LT 90
  REPEAT 2 [FD 240 RT 90 FD 100 RT 90]
  RT 90 FD 100 LT 90
  REPEAT 2 [FD 120 RT 90 FD 90 RT 90]
  RT 90 FD 90 LT 90
  REPEAT 2 [FD 220 RT 90 FD 60 RT 90]
  FD 220 RT 90 REPEAT 3 [FD 60 LT 120]
  LT 90 BK 220
END
```

**April 18, 2023**

**Challenge No. 29**

Pizza, anyone? Try drawing pizzas cut into 6, 8, and 12 slices.



Our solutions:

**Pizza for 6:**

```
STAMPOVAL 100 100  
REPEAT 6 [FD 100 BK 100 RT 60]
```

**Pizza for 8:**

```
STAMPOVAL 150 150  
REPEAT 8 [FD 150 BK 150 RT 45]
```


**Pizza for 12:**

```
STAMPOVAL 175 175  
REPEAT 12 [FD 175 BK 175 RT 30]
```

April 25, 2023

## Challenge No. 30

Can you draw concentric squares in Logo?  
After you write that code, see if you can simplify it using a procedure with inputs.



Our solutions:

```
TO SQUARES
  DRAW SETW 2
  REPEAT 4 [FD 30 RT 90]
  PU SETXY [-30 -30] PD
  REPEAT 4 [FD 90 RT 90]
  PU SETXY [-60 -60] PD
  REPEAT 4 [FD 150 RT 90]
  PU SETXY [-90 -90] PD
  REPEAT 4 [FD 210 RT 90]
  PU SETXY [-120 -120] PD
  REPEAT 4 [FD 270 RT 90]
  PU SETXY [-150 -150] PD
  REPEAT 4 [FD 330 RT 90]
  PU SETXY [-180 -180] PD
  REPEAT 4 [FD 390 RT 90]
  PU SETXY [-210 -210] PD
  REPEAT 4 [FD 450 RT 90]
  PU SETXY [-240 -240] PD
  REPEAT 4 [FD 510 RT 90]
  PU SETXY [-270 -270] PD
  REPEAT 4 [FD 570 RT 90]
END
```

```
TO EASYSQUARES :SIZE :POS
  IF :SIZE > 600 [STOP]
  REPEAT 4 [FD :SIZE RT 90]
  PU SETXY LIST :POS :POS PD
  EASYSQUARES (:SIZE + 60) (:POS - 30)
END

EASYSQUARES 30 -30
```

A more flexible procedure:

```
TO FLEXSQUARES :SIZE :POS :INC
  IF :SIZE > 600 [STOP]
  PU SETXY LIST :POS :POS PD
  REPEAT 4 [FD :SIZE RT 90]
  FLEXSQUARES (:SIZE + (:INC * 2))
    (:POS - :INC) :INC
END

FLEXSQUARES 10 10 10
FLEXSQUARES 20 15 5
FLEXSQUARES 20 30 40
```